

# HP 3421A Data Acquisition/Control Unit



**Programming the HP 3421A  
with the HP-71B and the HP-75D**

**Product Note 3421A-2**



# Table of Contents

---

## I. Introduction

Overview

## II. Programming the HP 3421A with the HP-71B

HP-71B Overview

HP-IL Basics

Programming Basics

Printers (HP 82162A, HP 2225B)

Mass Storage (HP 82161A, HP 9114A)

Data Logging

Example Programs

## III. Programming the HP 3421A with the HP-75D

HP-75D Overview

HP-IL Basics

Programming Basics

Printers (HP 82162A, HP 2225B)

Mass Storage (HP 82161A, HP 9114A)

Data Logging

Example Programs

## IV. Appendix

HP 3421A Programming Codes

HP-IL Accessories

Manual Part Numbers

Further Information on Mass Storage Devices

Other Data Acquisition Products



\*HP-IL (Hewlett-Packard Interface Loop): HP's new serial interface designed for small, low cost measurement systems.

\*\*HP-IB (Hewlett-Packard Interface Bus): Not just IEEE-488, but the hardware, documentation and support that delivers the shortest path to a measurement system.

# I. Introduction

This product note has been designed as an introduction to programming the HP 3421A with either the HP-71B or the HP-75D. The second section is on programming the HP 3421A with the HP-71B, and the third section is on programming the HP 3421A with the HP-75D. HP-IL printers and mass storage devices are also described within each section, and the HP 3421A programming codes and other HP-IL accessories are listed in the Appendix.

## Overview

An affordable, easy to use system is now available to solve your data acquisition applications where you need to measure DC and AC voltage, temperature, frequency, and resistance; or read and write digital information and actuate control signals. The heart of the system is the HP 3421A Data Acquisition/Control Unit that provides low cost measurement and control for your bench or portable test applications. The HP 3421A can be programmed via HP-IL by either the HP-71B or the HP-75D, which are both excellent low cost portable controllers. There are several HP-IL printers and mass storage devices available to provide hard copies and storage of the results of your data acquisition measurements. The data acquisition systems discussed in this product note include the following:

- Controllers:** HP-71B (with the 82401A HP-IL Module)  
HP-75D (with the 00075-15001 I/O ROM)
- Printers:** HP 2225B, 82162A
- Mass Storage:** HP 9114A, 82161A
- Data Acquisition Unit:** HP 3421A
- HP-IL Cables:** HP 10833A/B/C/D (1,2,4,0.5 metre)  
HP 11340A/B/C (20,50,100 metre)

## The HP-71B

The HP-71B is a powerful handheld computer that comes standard with:

- 17.5 Kbytes of user memory
- powerful BASIC language (including statistics)
- timer and clock
- battery operation (continuous memory)
- user definable keyboard
- CALC mode for powerful calculations

The HP-71B has the most commonly used BASIC commands like IF, THEN, ELSE, FOR, NEXT, etc. as dedicated keys to aid programming. There is also a separate numeric keypad for evaluating equations in its CALC mode. The addition of the HP 82401A HP-IL Module allows the HP-71B to program the HP 3421A with commands like OUTPUT and ENTER which greatly simplify programming and documentation.

## The HP-75D\*

- The HP-75D handheld computer comes standard with:
- 16 Kbytes of user memory
  - powerful BASIC language
  - timer and clock
  - battery operation (continuous memory)
  - built-in magnetic card reader
  - built-in bar code reader interface
  - built-in HP-IL

\*The instructions and programs for the HP-75D in this product note also apply to the HP-75C (with I/O ROM).

The addition of the HP-75D I/O ROM (part number 00075-15001) gives the HP-75D the capability of controlling the HP 3421A with OUTPUT and ENTER statements. The HP-75D also features a large keyboard for entering programs.

## The HP 2225B And HP 9114A

Two new HP-IL peripherals that feature high performance at a relatively low cost are the HP 9114A disc drive and the HP 2225B printer. The HP 9114A is a double sided 3 1/2 inch disc drive that features battery operation and is capable of storing 630 Kbytes of programs and data. The HP 2225B is an 80 column printer that has a disposable print head, battery operation and graphics capability.

## The HP 3421A

The HP 3421A measures voltage and resistance using a built-in integrating A/D converter with 3 1/2 - 5 1/2 digits of selectable resolution. Temperature measurements may be made with thermocouples (type J, K, T, E, R, or S), thermistors or RTD's. Transducer frequency outputs to 10 kHz may be measured with the HP 3421A's built-in counter. These quantities are measurable either from the front terminals or through up to three 10 channel scanner assemblies. Digital I/O and breadboard assemblies are also available for sensing, control or custom applications.

## Battery

Internal battery power lets the HP 3421A go wherever the measurement requires. For remote data logging applications you can combine the 3421A with either the HP-71B or the HP-75D. The HP 11340A/B/C HP-IL cables allow you to control the HP 3421A from 20, 50 or even 100 metres away.

## Calibration

Calibration of the HP 3421A is simplified by eliminating all internal adjustments. Just connect your standards and store calibration constants in non-volatile memory. Calibration is performed using either the HP-IL or the optional HP-IB interface. Complete calibration without adjustments or removal of the HP 3421A's covers lowers your cost of ownership.

## Measurement Functions

- DCV:** 300mV - 300V full scale; 1 $\mu$ V sensitivity (300 mV rge).
- ACV:** 3V - 30V full scale; 100 $\mu$ V sensitivity (3 V rge).
- Resistance:** 2 or 4-wire configuration; 300 $\Omega$  - 30M $\Omega$  full scale; 1m $\Omega$  sensitivity (max).
- Temperature:**
  - Thermocouple Types - T (linearization included in 3421A)
  - J, K, T, E, R, S, (reference junction compensation)
- Frequency:** 1 Hz - 10 kHz frequency range; totalize to 65,535 counts.

## Optional Assemblies

- Multiplexer Assembly:** 10 channel, 2-wire scanner. Up to 2 channels per assembly may be configured as actuator switches (252 VAC, 2 Amp max). Assembly includes thermocouple reference junction compensation.
- Digital I/O Assembly:** 8 input and 8 output lines. Inputs are optically isolated; 24V and 25mA max. Outputs are isolated open collector; 42V and 300mA max.
- Breadboard Assembly:** Allows user-designed circuits for custom functions.



Here are some examples of HP-71B statements that use these device specifiers:

- The `D=DEVADDR("HP3421A")` statement will set D equal to the address of the HP 3421A which is identified by its device ID of HP3421A.
- The `PRINTER IS:PRINTER` statement will assign the first printer class device it finds as the "printer" device.
- The `COPY filename TO filename:TAPE` statement finds and identifies the first mass storage class device in the loop and then copies the file to the mass storage device.

If you have more than one of these devices, you can specify the device you want to use by its address in the loop. For example, if you had two printers, two mass storage devices, two data acquisition units; the device specifiers would be:

First Device	Second Device
"HP3421A(1)"	"HP3421A(2)"
:PRINTER(1)	:PRINTER(2)
:TAPE(1)	:TAPE(2)

See the HP 82401A manual for more information.

## Programming Basics

### Displaying Any Program Line (`[A]`, `[V]`, `[G A]`, `[G V]`)

When you are entering your program you can use the `[A]` key (or `[V]` key) to bring the line preceding (or following) the current line to the display for editing. The `[G A]` and `[G V]` keys bring the lowest or highest numbered line to the display, ready to edit.

### Editing Any Line (`[F BACK]`, `[F -CHAR]`, `[F I/R]`, `[F -LINE]`)

- `[F BACK]` key backspaces the cursor one position and erases the character at that position.
- `[F -CHAR]` key erases the character at the cursor and moves the character left one space to fill in the gap.
- `[F I/R]` key switches between the Replace cursor  $\blacksquare$  and the Insert cursor  $\blacktriangleright$ . The Insert cursor allows you to insert characters to the left of the arrow.
- `[F -LINE]` erases all characters starting from the character at the cursor through the right end of the line.

## Developing, Editing and Saving a Program

Before typing the program lines listed below, make sure that you are in the BASIC mode (`>` prompt), instead of the CALC mode (CALC in the far right side of the display). Let's look at an example program that we'll call DCV that will program the HP 3421A to measure DC volts on a specified channel, and to then display the results on the HP-71B.

To create a new BASIC file, use the EDIT filename command by typing:

`EDIT DCV` and pressing `[ENDLINE]`

You can also have the HP-71B automatically number the program lines for you by typing:

`AUTO` and pressing `[ENDLINE]`

You have just created a new BASIC file called DCV. Now enter the following program lines (remember to press `[ENDLINE]` after you type each program line into the display).

	Comments
10 I DCV HP 71/3421A	Name of the program
20 REAL V	V is a real variable
30 CLEAR	Clear HP-IL
40 D=DEVADDR("HP3421A")	Assign D = HP 3421A address
50 OUTPUT :D ; "DCV5"	Measure DCV on channel 5
60 ENTER :D :V	Put results in variable V
70 DISP V	Show results on HP-71B
80 END	End of the program

### Listing 1

The `D=DEVADDR("HP3421A")` command sets D to the address of the HP 3421A (device ID of HP3421A). Remember that there are a number of different ways to identify a particular device. Setting D as the address of the HP 3421A is arbitrary, you could have used another variable such as C or X etc.

Now you are ready to run this program. Press the `[ATTN]` key to clear the HP-71B's display and type `RUN DCV` (or press the `[RUN]` key). If you have entered the program correctly, the HP 3421A will close channel 5, measure DC volts and then display the results on the HP-71B.

When you program the HP 3421A with the DCV command, the HP 3421A A/D converter autoranges, compensates for internal offsets (autozero), sets the A/D converter to the 5 1/2 digit mode (17 bits), and makes a precision DC voltage measurement. See the Appendix for more information on HP 3421A commands.

This program can be easily modified to measure and display **AC voltage, resistance, temperature, or frequency** on any of the appropriate channels that you have in your HP 3421A. All you have to do is change the program codes in line 50 (see the Appendix of this product note for a list of HP 3421A programming codes). For example, you could change line 50 to one of the following lines to program the HP 3421A for a different measurement:

Alternatives for line 50	Comments
50 OUTPUT :D;"ACV2"	Measure AC voltage on channel 2
50 OUTPUT :D;"TEM7"	Measure thermocouple, channel 7
50 OUTPUT :D;"F3CLS6"	Two wire ohms(F3),channel 6(CLS6)
50 OUTPUT :D;"F1N3CLS9"	DC volts, 3 1/2 digits, channel 9

The program DCV can also be modified to take readings on a number of channels and display the results. To edit the existing file DCV you can simply use the `[A]` key and `[V]` key to reach the appropriate line. If you had several program files in the HP-71B and the current file is not DCV, then execute a `EDIT DCV` statement (type `EDIT DCV` and press `[ENDLINE]`).

Let's change the program DCV to measure DC Volts on channels 3 through 7. The program changes required to do this are shown below in the program **Listing 2**. Change the program lines 20, 50, 55, 60, 70 and 75 to the following:

	Comments
10 I DCV HP 71/3421A	Name of the program
20 DIM V(5)	Dimension the array V
30 CLEAR	Clear HP-IL
40 D=DEVADDR("HP3421A")	Assign D = HP 3421A address
50 OUTPUT :D ; "DCV3-7"	Measure DCV on channel 3-7
55 FOR I=1 TO 5	Set up FOR-NEXT loop
60 ENTER :D :V(I)	Put the results in the array V
70 DISP V(I)	Display the readings
75 NEXT I	Increment I
80 END	End of the program

### Listing 2

You can bring line 50 (or any other line) to the HP-71B's display by pressing the `[V]` key until line 50 is on the display, or by executing a `FETCH 50` command. Once line 50 is displayed, you can make the changes shown above by using the `[<]` or `[>]` keys to position the cursor at the proper character. Use the `[I/R]` key to toggle between the insert (`[I]`) and the replace (`[R]`) mode.

To add lines 55, 75, press the `[ATTN]` key (`[ON]` key) to clear the display, enter 55 or 75 and the program line, and press `[ENDLINE]`. Remember to use the typing aids which have already defined keys for the FOR (`[F/R]`), TO (`[F/T]`) and NEXT (`[F/Y]`) statements. If you need more information about editing an existing program, see section 1 or section 8 of the HP-71B Owner's manual.

You can now run the program DCV again (press `[RUN]`). If you have edited the program correctly, the HP 3421A should measure DC voltage on channels 3 through 7 and display the results on the HP-71B.

Another statement that you should add to this program is the `OPTION BASE 1` statement which sets the array subscript to start at 1 (instead of 0). Enter line 15 as shown:

### 15 OPTION BASE 1

You can have the HP-71B renumber the program lines starting at the beginning of the program by executing the `RENUMBER` statement (press `[ATTN]`, type `RENUMBER` and press `[ENDLINE]`). Once you have done this, the new program should look like **Listing 3**:

```
10 | DCV HP 71/3421A
20 OPTION BASE 1
30 DIM V(5)
40 CLEAR
50 D=DEVADDR("HP3421A")
60 OUTPUT :D : "DCV3-7"
70 FOR I=1 TO 5
80 ENTER :D :V(I)
90 DISP V(I)
100 NEXT I
110 END
```

### Listing 3

The HP-71B has saved this program under the filename DCV in its internal memory. To create another program you simply execute another `EDIT` filename statement. See section 6 of the HP-71B owner's manual for information on file operations.

This program can be further modified to program the HP 3421A to take other multi-channel measurements by changing the `OUTPUT` statement in line 60 (see the Appendix of this product note for a summary of these codes). You may also have to redimension the array V in line 30 and the `FOR-TO` statement in line 70 for a different number of channel measurements. For example, to read AC voltage on channels 2-9, you would change lines 30, 60, and 70 in **Listing 3** to the following:

```
30 DIM V(8)
60 OUTPUT :D : "ACV2-9"
70 FOR I=1 to 8
```

## Printers (HP 82162A, HP 2225B)



The addition of an HP-IL printer like the HP 82162A or HP 2225B will allow you to obtain a hard copy of your program, or to print out the results of your measurements. You can assign the first printer device in the loop to be the program printer with a `PRINTER IS :PRINTER` statement. If you turn the printer (HP 2225B or 82162A) on before you turn on the HP-71B, the HP-71B should automatically assign the first printer device on HP-IL as the system printer. There are other ways of assigning the system printer such as `PRINTER IS:HP 2225B` (see the HP 82401A manual).

To verify that your printer is working properly, press the `[ATTN]` key to clear the HP-71B's display, type `PRINTER IS :PRINTER` and press `[ENDLINE]`. After you have done this, type `PRINT "TEST"` and press `[ENDLINE]`. The printer should now have printed out the word TEST. See the HP 82401A manual for information on the `PRINTER IS` statement.

Let's add one more line to our program DCV and change line 90 in **Listing 3** from a `DISP` statement to a `PRINT` statement. Remember, to add or change program lines, execute an `EDIT DCV` statement, then a `FETCH 25` (or 90) statement.

```
25 PRINTER IS:PRINTER
90 PRINT "CHANNEL";I+2;" = ";V(I)
```

After you have added lines 25 and 90 to the program DCV, you can renumber the program by typing `RENUMBER` and pressing `[ENDLINE]`. You can also obtain a program listing on your printer by typing `PLIST` and pressing `[ENDLINE]`. If you have edited the program DCV correctly, the measurements taken on the HP 3421A should be printed (along with the channel number) on your HP 2225B (or HP 82162A) printer. A program listing and sample print out (from the HP 2225B) is shown on the next page in **Listing 4**.

```

10 I DCV HP 71/3421A
20 OPTION BASE 1
30 PRINTER IS :PRINTER
40 DIM V(5)
50 CLEAR
60 D=DEVADDR("HP3421A")
70 OUTPUT :D : "DCV3-7"
80 FOR I=1 TO 5
90 ENTER :D :V(I)
100 PRINT "CHANNEL " :I+2:"=" :V(I)
110 NEXT I
120 END

```

```

CHANNEL 3 = .032017
CHANNEL 4 = .067904
CHANNEL 5 = .094889
CHANNEL 6 = .121002
CHANNEL 7 = .135604

```

**Listing 4**

By adding a PRINT USING and an IMAGE statement you can format the output of the HP 82162A printer so that each pass will be on one line instead of two (See **Listing 4a**).

```

10 I DCV HP 71/3421A
20 OPTION BASE 1
30 PRINTER IS :PRINTER
40 DIM V(5)
50 CLEAR
60 D=DEVADDR("HP3421A")
70 OUTPUT :D : "DCV3-7"
80 FOR I=1 TO 5
90 ENTER :D :V(I)
100 PRINT USING :I+2,"=" :V(I)
110 NEXT I
120 IMAGE ("CHANNEL",DD,"=",DDD.DDDDDD)
130 END

```

```

CHANNEL 3 = .033251
CHANNEL 4 = .056545
CHANNEL 5 = .092755
CHANNEL 6 = .114905
CHANNEL 7 = .131887

```

**Listing 4a**

## Mass Storage (HP 82161A, 9114A)



For purposes of clarity, this product note will refer to the HP 82161A and the HP 9114A with the same device word (:TAPE). The mass storage commands (e.g. COPY, ASSIGN, INITIALIZE) executed on the HP-71B will work with either the HP 82161A or the HP 9114A. You can change the device word :TAPE to device word :MASSMEM for the HP 9114A disc drive (either one will work for the HP 9114A).

In order to use a mass storage device like the HP 82161A digital cassette drive or the HP 9114A disc drive, you must first initialize the medium in the device. To initialize the mass storage device, type **INITIALIZE :TAPE** and press **ENDLINE**. This will initialize the HP 82161A cassette for a default directory size of 128 file entries and the HP 9114A for a directory size of 616 file entries. The directory size that you specify may depend on whether you'll have many short files or a few long files. You can specify the number of file entries by typing: **INITIALIZE:TAPE, number of file entries**. You save programs on the mass storage device by typing: **COPY filename TO filename:TAPE**.

For example, you can save the file DCV by typing **COPY DCV TO DCV:TAPE** and pressing **ENDLINE**. To transfer a file from the mass storage device to the HP-71B, type **COPY DCV:TAPE TO DCV** and press **ENDLINE**. If you have two mass storage devices connected to the HP-71B, you can transfer a file from the first mass storage device to the second mass storage device by typing **COPY filename:TAPE(1) TO filename:TAPE(2)** and pressing **ENDLINE**.

To save data you first have to create the data file with: **CREATE DATA filename:TAPE,# of records, number of bytes per record**.

If you do not specify the number of bytes per record, it will default to 256 bytes per record.

Once you have created the data file, you can write data to that file by assigning a channel number to that file with a **ASSIGN # channel number TO filename**, and then executing a **PRINT # channel number** statement (see the example program STORE1 on the following page).

## Reading Storage

In many data acquisition applications you may need to know approximately how many HP 3421A readings the

HP-71B can store on the cassette (or disc). For storing data in a data file, the HP-71B:

- uses 8 bytes to store a numeric value (e.g. HP 3421A reading)
- uses 3 bytes plus 1 byte per character to store a string (e.g. current time of day)

The maximum number of HP 3421A readings that you can store on the cassette or disc is shown below (assuming only one file is used for all readings). See the Appendix for an explanation of how these values were determined.

	HP 82161A Cassette Drive	HP 9114A Disc Drive
Available Bytes	130,304	630,016
Maximum Number of Readings	16,288	78,752

Normally you will want to store the time the data was taken along with the actual data. Let's look at an example to calculate the typical number of HP 3421A readings you can store on the cassette (or disc). If you store the time (11 bytes) plus 10 HP 3421A readings (80 bytes), you will use 91 bytes to store 10 readings. You could store 14,319 readings (130,304 bytes/9.1 bytes per reading) on the HP 82161A cassette drive or 69,232 readings (630,016 bytes/9.1 bytes per reading) on the HP 9114A disc drive (assuming that all data is stored in one DATA file).

Let's write a program called STORE1 that takes DC voltage readings from channels 5 through 9 in the 5 1/2 digit mode and stores the results on tape(or disc). Type **EDIT STORE1** and press **[ENDLINE]**, type **AUTO** and press **[ENDLINE]**, and enter the following program in **Listing 5**.

	Comments
10 STORE1 HP 71/3421A	Name of the program
20 OPTION BASE 1 @ FIX 6	Array starts at 1; fix format at 6 decimal places
30 DIM V(5) @ CLEAR	Dimension array V; Clear HP-IL
40 CREATE DATA DATA1:TAPE,1,40	Create data file DATA1 with 1 record, 40 bytes/record
50 D=DEVADDR("HP3421A")	Set D to HP 3421A address
60 OUTPUT :D:"DCV5-9"	Measure DC volts, channels 5-9
70 FOR I=1 TO 5	FOR NEXT Loop to enter the data into the array V
80 ENTER :D :V(I)	
90 NEXT I	
100 ASSIGN #1 TO DATA1:TAPE	Assign I/O channel 1 in the HP-71B to data file
110 PRINT #1:V()	Sends data from array to file DATA1
120 ASSIGN #1 TO *	Close the file DATA1, release I/O channel 1
130 END	

**Listing 5**

The @ concatenation character in lines 20 and 30 joins statements, which enables you to enter more than one statement in a program line. See the HP-71B manuals for more information on the @ character.

The CREATE statement in line 40 creates a data file named DATA1 on the cassette (or disc) that is one record consisting of 40 bytes for that record. Since the HP-71B needs 8 bytes to store a noninteger, you need 40 bytes to store 5 DC voltage readings in DATA1. You do not have to specify the number of bytes per record (the default is 256 bytes per record), but make sure that the records have enough length (bytes) to store all your data.

The ASSIGN #1 assigns an I/O channel 1 to the file DATA1 on the mass storage device. An I/O channel (1-255) includes a buffer in main RAM that's used to transfer data to and from the file. The PRINT #1 sends the data from the array V to I/O channel 1 which sends it to file DATA1.

If you have properly connected the HP 3421A, 82161A or 9114A, and the 71B; and you have entered the program correctly, this program should create a data file, take 5 readings from the HP 3421A, and store the results in the file DATA1. Press **[RUN]** or type **RUN STORE1** and press **[ENDLINE]**.

If you try to run this program more than once, the HP-71B will inform you that you already have a file called DATA1. This is to protect your data from being erased by accident. To get around this, type **PURGE DATA1:TAPE** and press **[ENDLINE]**, or rename your data file.

For some data acquisition applications, you may want to record the data for a period of time and then analyze it later. Let's add the capability of reading the data in the file DATA1 from the mass storage device into the HP-71B and displaying a specific channel reading. Add the following lines to the program STORE1:

	Comments
112 RESTORE #1 @ DIM R(5)	Sets the file pointer to start the data; dimension R
114 READ #1:R()	Read DATA1 from the mass storage device to array R
116 DISP "CHANNEL 5=",V(1)	Display data before storage
117 WAIT 1	
118 DISP "FROM DATA FILE, CHANNEL 5=",R(1)	Display data after storage

**Listing 5a**

Line 116 displays the reading on channel 5 prior to storing it on the mass storage device, while line 118 displays the reading that has been stored on the cassette (or disc). This is a quick check to verify that the data has been stored properly. You can modify the program further to verify all the readings have been correctly stored.

Save this program on your mass storage device by typing: **COPY STORE1 TO STORE1:TAPE** and pressing **[ENDLINE]**.

## File Catalog


You can find out which files you have in the HP-71B's memory by typing **CAT ALL** and pressing **[ENDLINE]**, and then using the **[V]** key to display the next file. You can also find out which files are on your mass storage device by typing **CAT :TAPE** and pressing **[ENDLINE]**, and again using the **[V]** key. To get a print out of these catalogs, type **DISPLAY IS:PRINTER** and press **[ENDLINE]**, execute **CAT ALL** or **CAT :TAPE** and use the **[V]** key. You can get back to normal display operation by typing **DISPLAY IS \*** (or **DISPLAY IS:DISPLAY**) and pressing **[ENDLINE]**.



# Data Logging

Data logging consists of a sequence of measurements (e.g. temperature) repeated at specific time intervals and either saved on a mass storage device, a printer or both. The combination of a HP 3421A, 71B, 2225B, and either the HP 82161A or the HP 9114A is a powerful data logger that is well suited for portable or nonportable applications.

Let's start out with a simple example to understand the structure of a data logging program, and then modify the program to see how you could adapt this program to fit your application. This program assumes that you have a HP-71B (with 82401A), an HP 3421A (with an option 020), an HP 2225B (or HP 82162A printer) and an HP 9114A (or an HP 82161A) mass storage device. This first example program will be called LOG1, so type **EDIT LOG1**, press **[ENDLINE]**, and enter in the program lines (this program assumes that you have initialized the storage media).



## Comments

```

10 | LOG1 HP 71/3421A ----- Name of the program
20 OPTION BASE 1 @ FIX 6 ----- Array subscript, fix digits
30 DIM V(8),T$(8) @ CLEAR @ T1=1 ----- Dimension variables,clear loop, set counter T1
40 CREATE DATA DATALOG1:TAPE,5,75 ----- Create DATALOG1, 5 records, 75 bytes per record
50 PRINTER IS :PRINTER ----- Set up printer
60 D=DEVADDR("HP3421A") ----- Set D = address of 3421A
70 ASSIGN #1 TO DATALOG1:TAPE ----- Set up I/O channel to the file DATALOG1 on mass storage device
80 ON TIMER #1,60 GOTO 90 ----- Timer interrupt every 60 seconds, go to line 90 on interrupt
90 T$=TIME$ ----- Read the current time
100 OUTPUT :D : "DCV2-5" ----- DC volts, channels 2-9
110 FOR I=1 TO 8 -----
120 ENTER :D :V(I) ----- Put measurements in array V
130 NEXT I -----
140 PRINT #1:T$,V(I) ----- Transfer measurements from T$,V to DATALOG1 file on tape (disc)
150 PRINT " " ----- Print out a blank line
160 PRINT "TIME=":T$ ----- Print out the time
170 FOR I=1 TO 8 -----
180 PRINT USING 250:I+1,V(I) ----- Print out the results
190 NEXT I -----
200 T1=T1+1 ----- Increment counter T1
210 IF T1=6 THEN 'EXIT' ----- After 5 passes, exit program
220 GOTO 220 ----- Wait for next timer interrupt, go to line 90 on interrupt
230 'EXIT': OFF TIMER #1 ----- Disable timer interrupt
240 ASSIGN #1 TO * ----- Close I/O channel, file DATALOG1
250 IMAGE ("CHANNEL",DD," =",DDD.DDDDDD) ----- Format statement for PRINT USING (line 180)
260 END

```

```

TIME=07:21:45
CHANNEL 2 = .092701
CHANNEL 3 = .157028
CHANNEL 4 = .200809
CHANNEL 5 = .254113
CHANNEL 6 = .288591
CHANNEL 7 = .370710
CHANNEL 8 = .394410
CHANNEL 9 = .465850
:
TIME=07:25:45
CHANNEL 2 = .088859
CHANNEL 3 = .198262
CHANNEL 4 = .282461
CHANNEL 5 = .403620
CHANNEL 6 = .467530
CHANNEL 7 = .536610
CHANNEL 8 = .604020
CHANNEL 9 = .674420

```

Listing 6

The CREATE DATA statement in line 40 of **Listing 6** creates the data file DATALOG1 with 5 records of 75 bytes each. Each pass will require 75 bytes (8 readings x 8 bytes per reading plus 11 bytes for the time). You could modify line 40 to create the data file with 1 record of 375 bytes:

```
CREATE DATA DATALOG1:TAPE,1,375
```

or 2 records with the default condition of 256 bytes each

```
CREATE DATA DATALOG1:TAPE,2
```

and the program would still run properly. If the data file you create only had 1 record of 256 bytes, the program will halt during the fourth pass to indicate an end of file error.

This program (**Listing 6**) will measure DC voltage on channels 2 through 9, store the time and results on tape in data file DATALOG1, and print out the results. The program will wait for 1 minute after the start of the previous pass and then start another pass. After 5 passes (i.e. iterations), the program will stop. Remember, if you want to run this program more than once, either change the name of the data file in line 40 (and 70), or purge the old data file DATALOG1 (type `PURGE DATALOG1:TAPE` and press `ENDLINE`).

Now that you have recorded the data, you can access the data in the file DATALOG1 on the tape at a later time. The program in **Listing 7** is an example of how to read the entire file DATALOG1 and print out the results from pass number 1 and pass number 5 in a tabular format. There are many different ways you can format the results with `PRINT USING` and `IMAGE` statements. See the HP-71B manual for more information. Below is the program listing and a sample printout (using the HP 2225B printer).

This program can be modified to read and print out other selected passes and readings, or you could perform some statistics on the data with the HP-71B (see the example programs on the following pages).

The data logger program (LOG1) can be easily modified for different measurement functions, different time intervals and

number of iterations (or passes). You can modify the program to either print or record only, instead of doing both.

Before you modify LOG1, save it on your mass storage device with:

```
COPY LOG1 TO LOG1:TAPE and pressing ENDLINE.
```

Let's call the new program LOG2, and copy the existing program LOG1 by typing `COPY LOG1 TO LOG2` and pressing `ENDLINE`. Now you have saved the program LOG1 on tape, LOG1 is still in the HP-71B's memory, and you have created a file LOG2 that you can modify. Type `EDIT LOG2` and press `ENDLINE`. Change the program lines that are highlighted in **Listing 8**.

This program will measure DC voltage on channels 2-9, store the results on the mass storage device in a data file called DATALOG2, print out the results, and repeat the process every 5 minutes for 2 hours. The changes made to the program were creating a data file of the proper size (line 40), changing the timer to a different time between passes (line 80), and changing the number of passes (line 210).

Also note that line 220 changed to a `BYE` statement; this turns the HP-71B off. When the timer has elapsed its 5 minutes, it will turn the HP-71B back on and proceed with the program. When the HP-71B is turned off, it will power down any HP-IL peripherals that will respond to a loop power down command (HP 3421A\*, 82161A, 82162A). When the HP-71B wakes up and sends out a command to the peripherals, it will reawaken them.

The data file DATALOG2 has been created in line 40 with 24 records; since the number of bytes per record has not been specified, it defaults to 256 bytes per record.

*\*NOTE: Current HP 3421A's do not turn off their display when entering a loop power down state. HP 3421A's with serial numbers below 2238A03053 turn off their display when they receive a power down command.*

```
10 | RLOG1 HP 71/3421A
20 PRINTER IS :PRINTER
30 OPTION BASE 1 @ FIX 6
40 DIM R1(8),R2(8),R3(8),R4(8),R5(8),T1$,T2$,T3$,T4$,T5$
50 ASSIGN #1 TO DATALOG1:TAPE
60 READ #1:T1$,R1(),T2$,R2(),T3$,R3(),T4$,R4(),T5$,R5()
70 ASSIGN #1 TO *
80 PRINT "START TIME=",T1$,"END TIME=",T5$
90 PRINT USING 130:"2","3","4","5","6","7","8","9"
100 PRINT USING 140
110 PRINT USING 150:R1(1),R1(2),R1(3),R1(4),R1(5),R1(6),R1(7),R1(8)
120 PRINT USING 150:R5(1),R5(2),R5(3),R5(4),R5(5),R5(6),R5(7),R5(8)
130 IMAGE 8(2X,"CHAN#",A,X)
140 IMAGE 8(X,7,"-",X)
150 IMAGE 8(D.DDDDDD,X)
160 END
```

START TIME=		07:21:45			END TIME=		07:25:45	
CHAN#2	CHAN#3	CHAN#4	CHAN#5	CHAN#6	CHAN#7	CHAN#8	CHAN#9	
.092701	.157028	.200809	.254113	.288591	.370710	.394410	.465850	
.088859	.198262	.282451	.403520	.467530	.536610	.604020	.674420	

### Listing 7

## Comments

```

10 LOG2 HP 71/3421A ----- Name of the program
20 OPTION BASE 1 @ FIX 6 ----- Array subscript, fix digits
30 DIM V(8),T$(8) @ CLEAR @ T1=1 ----- Dimension variables, clear loop
40 CREATE DATA DATALOG2:TAPE,24 ----- Create DATALOG2, 24 records, 256 bytes per record (default)
50 PRINTER IS :PRINTER ----- Set up printer
60 D=DEVADDR("HP3421A") ----- Set D= address of 3421A
70 ASSIGN #1 TO DATALOG2:TAPE ----- Set up I/O channel to DATALOG2
80 ON TIMER #1,300 GOTO 90 ----- Timer interrupt every 5 minutes
90 T$=TIME$ ----- Read the current time
100 OUTPUT :D : "DCV2-9" ----- Program the 3421A to measure DC volts channel 2-9
110 FOR I=1 TO 8 -----
120 ENTER :D :V(I) ----- Put measurements in array V
130 NEXT I -----
140 PRINT #1,T1;T$,V() ----- Put measurements in DATALOG2; store pass 1 on record1, store pass 2 on record 2 etc
150 PRINT " " ----- Print out blank line
160 PRINT "TIME=";T$ ----- Print out the time
170 FOR I=1 TO 8 -----
180 PRINT USING 250:I+1,V(I) ----- Print out the results
190 NEXT I -----
200 T1=T1+1 ----- Increment counter T1
210 IF T1=25 THEN 'EXIT' ----- After 24 passes, exit program
220 BYE ----- Wait for next timer interrupt, turn HP-71B, 3421A etc. off, on interrupt go to line 90
230 'EXIT': OFF TIMER #1 ----- Disable timer interrupt
240 ASSIGN #1 TO * ----- Close I/O channel, file DATALOG2
250 IMAGE ("CHANNEL",DD," =",DDD.DDDDDD) ----- Format statement for PRINT USING (line 180)
260 END

```

## Listing 8

Line 140 changes from a PRINT #1;T\$,V() statement to a PRINT #1,T1;T\$,V() statement; this will store each pass on a different record (e.g. pass number 15 is stored on record 15 in DATALOG2). This means that when you want to read the data off the mass storage device, you can specify a particular record (pass) to be transferred to the HP-71B (this is called random access as opposed to sequential access).

**Listing 9** is a program called RLOG2 that will read pass number 12 from the file DATALOG2 on the mass storage device, and print out the results.

```

10 RLOG2 HP 71/3421A
20 OPTION BASE 1 @ FIX 6
30 DIM V(8),T$(8) @ CLEAR @ PRINTER IS :PRINTER
40 ASSIGN #1 TO DATALOG2:TAPE
50 READ #1,12:T$,V()
60 PRINT "TIME=";T$
70 FOR I=1 TO 8
80 PRINT USING 110:I+1,V(I)
90 NEXT I
100 ASSIGN #1 TO *
110 IMAGE ("CHANNEL",DD," =",DDD.DDDDDD)
120 END

```

## Listing 9

Using these example data logger programs as a base, you should be able to modify them to fit most of your data logging applications. You can also add lines to the program to take an action based on a limit on a channel being exceeded (see the example programs).

## Thermistor, Thermocouple and RTD Linearization

Thermistor and thermocouple linearization programs are found on pages 76 and 78 of the HP 3421A manual; you should be able to run these programs by changing the OUTPUT 901 and ENTER 901 statements to OUTPUT :D and ENTER :D and adding an D=DEVADDR ("HP 3421A") statement to the beginning of the program. Linearization for a 100 Ohm Platinum RTD ( $\alpha=0.00385$ ) is shown in this example:

```

10 D=DEVADDR("HP 3421A")
20 OUTPUT :D : "FW02"
30 ENTER :D ; C
40 W=C/100
50 IF W<1 THEN 100
60 P1=3367.82144088
70 P2=13065764.8633
80 P3=-1723543.60565
90 P=P1-SQR(P2+P3*W) @ GOTO 150
100 P0=-241.996759172
110 P1=222.560617915
120 P2=25.2488238815
130 P3=-5.81268262546
140 P=P0+W*(P1+W*(P2+W*P3))
150 DISP P;"DEGREES C"
160 END

```

---

## Example Programs

By now you should have an understanding of how to program the HP 3421A with the HP-71B, and how to communicate with other HP-IL peripherals. The following programs are further examples of how to program the HP 3421A with HP-71B.

**Problem:** Program the HP 3421A to take 16 readings of DC voltage on channels 6-9 (4 readings per channel) and print the mean and standard deviation of those 16 readings.

**Solution:**

	<b>Comments</b>
10 OPTION BASE 1 @ STAT S(1) @ CLSTAT	Array subscript = 1, set up statistical array S
20 FIX 6 @ DIM A(4,4)	Fix digits, dimension array A,
30 PRINTER IS :PRINTER	Set up printer
40 D=DEVADDR("HP3421A")	Set D = address of 3421A
50 FOR I=1 TO 4	Set up FOR-NEXT loop
60 OUTPUT :D : "DCV6-9"	DC volts on channels 6-9
70 FOR J=1 TO 4	Set up loop to read data
80 ENTER :D : A(I,J)	Enter data into array A
90 ADD A(I,J)	Add data points to statistical Array S
100 NEXT J	
110 NEXT I	
120 PRINT "MEAN=" : MEAN(1)	Print mean of array S
130 PRINT "STD. DEV.=" : SDEV(1)	Print standard deviation of S
140 END	
MEAN= 1.815074	
STD. DEV.= 0.038717	

### Listing 10

**Problem:** You have two HP 3421A's in the loop, and you need to program the first one to measure DC volts on channels 3-7 and print out the results. The second HP 3421A needs to monitor temperature on channels 2-9 every 10 seconds for 10 minutes, and close actuator channel 1 for 30 seconds (turn on a fan), if any channel exceeds 30 degrees C.

**Solution:**

	<b>Comments</b>
10 OPTION BASE 1 @ FIX 6	Set array subscript; fix digits
20 DIM X(5),V(8) @ PRINTER IS :PRINTER	Dimension arrays X,V
30 D1=DEVADDR("HP3421A(1)")	Set D1 = address of first HP 3421A in the HP-IL loop
40 D2=DEVADDR("HP3421A(2)")	Set D2 = address of second HP 3421A in the HP-IL loop
50 PRINT "DATA FROM FIRST HP 3421A"	Print Message
60 OUTPUT :D1 : "DCV3-7"	DC volts, channels 3-7
70 FOR I=1 TO 5	Set up FOR-NEXT loop
80 ENTER :D1 : X(I)	Enter data into array X
90 PRINT "CHANNEL " : I+2 : " = " : X(I)	Print out the results
100 NEXT I	
110 ! PROGRAM SECOND 3421A	Comment line
120 PRINT " "	Print out a blank line
130 PRINT "DATA FROM SECOND HP 3421A"	Print message
140 ON TIMER #1,600 GOTO 'EXIT'	Exit program after 10 minutes
150 OUTPUT :D2 : "TEM2-9"	T type T/C's, channels 2-9
160 FOR I=1 TO 8	Set up FOR-NEXT loop
170 ENTER :D2 : V(I)	Enter data into array V
180 IF V(I)>30 THEN 'FAN'	If the temperature is greater than 30, goto fan routine
190 NEXT I	
200 WAIT 10 @ GOTO 150	Wait 10 seconds; return to monitoring temperature
210 'FAN': PRINT "CHANNEL " : I+1 : " > 30 DEG.C"	Print channel number > 30 deg.
220 OUTPUT :D2 : "CLS1"	Close actuator channel 1
230 WAIT 30	Wait 30 seconds
240 OUTPUT :D2 : "OPN1"	Open actuator channel 1
250 GOTO 150	Return to monitoring temp.
260 'EXIT': OFF TIMER #1	Turn off timer, exit program
270 END	

### Listing 11

**Problem:** You need to benchmark the HP-71B/3421A combination for reading rates (with and without transfer times) for various modes of operation.

**Solution:**

```

10 PRINTER IS :PRINTER _____ Set up printer
20 DIM A(30) _____ Dimension array A
30 D=DEVADDR("HP3421A") _____ D = HP 3421A address
40 OUTPUT :D ; "T0LS2-9,2-9,2-9,2-7:F1N3R0Z0" _____
50 T1=TIME _____ Read the current time
60 OUTPUT :D ; "T3" _____ Trigger the HP 3421A
70 T2=TIME _____ Read the time after
80 FOR I=1 TO 30 _____ measurements have been
90 ENTER :D ; A(I) _____ completed
100 NEXT I _____ Set up FOR-NEXT loop to
110 T3=TIME _____ transfer the data from the
120 PRINT "READING TIME=",30/(T2-T1);"RDGS/SEC" _____ HP 3421A's internal buffer
130 PRINT "READING PLUS TRANSFER TIME=",30/(T3-T1);"RDGS/SEC" _____ to the HP-71B.
140 END _____ Read the time after the
READING TIME= 25.000000 RDGS/SEC _____ measurement and transfer is
READING PLUS TRANSFER TIME= 8.902077 RDGS/SEC _____ completed

```

**Comments**

T0 Wait for trigger;enable channel list,  
reading storage  
LS2-9,2-9,2-9,2-7 Load channel list  
F1 DC volts  
N3 3 1/2 digits  
R0 Fixed range (3 volt)  
Z0 Auto Zero off

**Listing 12**

You can modify this program for a different number of digits of resolution, auto zero on or off, different function, or different range by changing the HP 3421A programming codes in line 40. This program will take readings on 30 channels; to take 30 readings on a single channel, change line 40 to the following: `40 OUTPUT :D;"T0LS4-4:F1N3R0Z0"`.

Here are some typical numbers for the HP 3421A controlled by the HP71B: (these numbers do not subtract the time it takes to read the HP-71B's internal clock, so the actual reading rates should be slightly faster)

DC Volts*		Auto Zero	
		OFF	ON
Different Channels	3 1/2 digits	25/8	21/8
	4 1/2 digits	17/7	12/6
	5 1/2 digits	3/3	2/1
Same Channel	3 1/2 digits	36/10	29/9
	4 1/2 digits	22/8	15/7
	5 1/2 digits	4/3	2/1

\* The first number is the reading rate (per second) into the HP 3421A's internal buffer; the second number is the reading rate (per second) including reading time plus transfer time to the HP-71B.

**Problem:** You need to set the HP 3421A's SRQ mask to enable the HP 3421A to interrupt the HP-71B when an abnormal condition exists (e.g. illegitimate HP 3421A command). When the HP-71B receives the interrupt, it will go to an interrupt handling sub-program and execute a serial poll command to the HP 3421A to determine the cause of the SRQ, and then display an error message.

**Solution:**

```

10 FIX 3 @ REAL V @ INTEGER P _____ Fix digits;declare P as an integer
20 D=DEVADDR("HP3421A") _____ Set D = address of HP 3421A
30 ON INTR GOSUB 'SRQ' _____ On interrupt, go to SRQ subroutine
40 ENABLE INTR 8 _____ Enable interrupt
50 OUTPUT :D ; "M32" _____ Set the HP 3421A's SRQ mask
60 OUTPUT :D ; "FR3" _____ Send an incorrect HP 3421A command
70 GOTO 70
80 END
90 |
100 'SRQ':
110 P=SPOLL(D) @ DISP "SPOLL=";P _____ Do serial poll on HP 3421A
120 DISP "ABNORMAL CONDITION" _____ Display message
130 END SUB

```

**Comments**

**Listing 13**

Serial Poll allows you to determine the current status of the HP 3421A. When the 3421A receives the Serial Poll message, it returns its Status Byte (from the Status Register). The value returned is the sum of the individual bits that are set. In this example, bits 6, 5 (and 0) are set, the value of P should be 97(64+32+1=97). The 8 bits of the Status Byte and their respective values are shown to the right.

	Bit 7 N/A Always	Bit 6 SRQ	Bit 5 Abnormal Condition	Bit 4 Low Battery	Bit 3 Event Occurred	Bit 2 Self Test	Bit 1 Power/on Reset	Bit 0 Data Ready
Decimal Value	128	64	32	16	8	4	2	1

Status Byte (Register)

# III. Programming The HP 3421A With The HP-75D

This section of the product note assumes that you have read sections 1 and 2 of the HP-75D Owner's manual and sections 1 and 5 of the HP-75D I/O ROM manual.

## HP-75D Overview



Pressing **ATTN** turns on the HP-75D, while pressing **SHIFT** **ATTN** turns it off. The **>** prompt in the HP-75D display is the BASIC prompt, showing that you are in the BASIC mode (as opposed to the TIME or APPT mode).

The HP-75D has a keyboard layout very similar to a typewriter to aid you in typing your programs. The **RTN** key will store the characters you've just typed into memory (a program, for example), or immediately executing the statement.

The **>** and **<** keys allow you to move the cursor back and forth along the line, while the **SHIFT** **>** and **SHIFT** **<** keys move the cursor immediately to either the first or last character of the line.

When your HP-75D is on, the **ATTN** (attention) key will clear the display when a program is not running, or suspend the program that it is currently running.

You can recall the last statement you typed by pressing **CTL** **FET** (hold the **CTL** key down while pressing **FET**). This recalls a temporary input buffer and displays the last statement for you. You can modify the statement and then execute the modified statement by pressing **RTN**.

When a program is running, the PRGM indicator is on in the bottom of the HP-75D's display. If you get an error message on the HP-75D, type **ERRN** and press **RTN**; the HP-75D will then display the error number which you can then look up in the HP-75D manuals for a further explanation.

### Setting the HP-75D Clock

You can set the time and date of the built in clock by pressing the **TIME** key, typing **SET** and pressing **RTN**. Type in the new date and time and press **RTN** or cancel this procedure by just pressing **RTN**. To get out of the TIME mode and back to the EDIT mode simply press the **EDIT** key. You can display the date and time from the EDIT mode by typing:

**DATE\$**           press **RTN**  
**TIME \$**           press **RTN**



## HP-IL Basics

Programming the HP 3421A with the HP-75D (using **OUTPUT**'s and **ENTER**'s) requires that the I/O ROM (part number 00075-15001) be installed in your HP-75D. Be sure the HP-75D is turned off before installing the I/O ROM, and before connecting or disconnecting interface cables. After you have installed the I/O ROM into the HP-75D, connect all of your HP-IL peripherals together in a continuous loop as described in your HP-75D I/O ROM manual. Turn on all the HP-IL peripherals before turning on the HP-75D.

The HP-75D I/O ROM manual describes how to distinguish HP-IL devices by their address (0 to 30) in the loop. The system controller (HP75D) assumes the address of 0 at power on, and then assigns the first device the address of 1, the second device in the loop is assigned the address of 2 etc. Once the HP-75D has assigned device addresses, you should then assign a device code to each device.

This section requires that you execute an **ASSIGN LOOP** statement that uses the accessory ID of the device to determine its class, then assigns a two-character device code to that device. If you type **AUTOLOOP ON** and press **RTN**, the HP-75D will automatically execute an **ASSIGN LOOP** statement when the HP-75D turns on.

This section will also assume for simplicity that the minimum configuration consists of an HP-75D (with the I/O ROM) and an HP 3421A, and that the maximum configuration consists of the HP-75D, 3421A, printer (either an HP 82162A or an HP 2225B), and a mass storage device (either an HP 82161A or an HP 9114A). If you have one of these configurations, and you have executed an **ASSIGN LOOP** (or **AUTOLOOP ON**) statement, the HP-75D should assign the following device codes:

		Device Code
Data Acquisition Unit	HP 3421A	:E1
Printer	HP 82162A	:P1
Printer	HP 2225B	:P1
Cassette Drive	HP 82161A	:M1
Disc Drive	HP 9114A	:M1

The E indicates the HP 3421A is in the Electronic Instrument device class, the P indicates the HP 82162A and HP 2225B are in the Printer device class, and the M indicates the HP 82161A and HP 9114A are in the Mass Storage device

\* **Caution:** The programs in this section assume that you have a relay multiplexer board in slot 0 of your HP 3421A. If you relay multiplexer (option 020, 021, or 022) is in another slot, use the appropriate channel number (see the HP 3421A manual). If you do not have a relay multiplexer, use DCV command without specifying a channel number, and the HP 3421A will read the DC voltage at the front terminals.

class. The 1 indicates that the device was the first one of its class found in the loop.

Here are some examples of HP-75D statements that use these device codes:

- The D=DEVADDR(':E1') statement will set D equal to the address of the HP 3421A which is identified by its device code :E1.
- The PRINTER IS ':P1' statement will assign the printer with device code as the "printer" device.
- The COPY 'filename' TO 'filename:M1' statement will copy a file to the mass storage device with the device code of :M1.

If you have more than one of these devices, you can specify the device you want to use by its address in the loop. For example, if you had two printers, two mass storage devices, two data acquisition units; the device codes would be:

	First Device	Second Device
3421A's	:E1	:E2
Printers	:P1	:P2
Mass Storage	:M1	:M2

See the HP-75D I/O ROM manual for more information.

## Programming Basics

### Displaying Any Program Line (▲, ▼, SHIFT ▲, SHIFT ▼)

When you are entering your program you can use the ▲ key (or ▼ key) to bring the line preceding (or following) the current line to the display for editing. The SHIFT ▲ and SHIFT ▼ keys bring the lowest or highest numbered line to the display, ready to edit.

### Editing Any Line (DEL, I/R, CLR, FET)

- DEL key erases the character at the cursor and moves the character left one space to fill in the gap. SHIFT DEL erases all the characters to the right of the cursor.
- I/R key switches between the Replace cursor (■) and the Insert cursor (⊕). The Insert cursor allows you to insert characters to the left of the arrow.
- CLR clears the display (but does not delete any characters)
- FET displays a FETCH command and waits for you to enter the line number of the current program to be displayed

## Developing, Editing and Saving a Program

Before keying in the program lines listed below, make sure that you are in the BASIC mode (➤ prompt), instead of the APPT mode. Let's look at an example program that we'll call DCV that will program the HP 3421A to measure DC volts on a specified channel, and to then display the results on the HP-75D.

To create a new BASIC file, use the EDIT filename command by typing:

EDIT 'DCV' and pressing RTN

You can also have the HP-75D automatically number the program lines for you by typing:

AUTO and pressing RTN

You have just created a new BASIC file called DCV. Now enter the following program lines (remember to press RTN after you type each program line into the display). These programs will only work if you have executed a ASSIGN LOOP statement (or an AUTOLOOP ON statement) as described in the previous page under HP-IL Basics.

	Comments
10 I DCV HP 75/3421A	Name of the program
20 REAL V	V is a real variable
30 CLEAR LOOP	Clear HP-IL
40 D=DEVADDR(':E1')	Assign D = HP 3421A address
50 OUTPUT D : "DCV5"	Measure DCV on channel 5
60 ENTER D : V	Put results in variable V
70 DISP V	Show results on HP-75D
80 END	End of the program

Listing 14

The D=DEVADDR(':E1') command sets D to the address of HP 3421A (device code ':E1'). Remember that there are a number of different ways to identify a particular device. Setting D as the address of the HP 3421A is arbitrary, you could have used another variable such as C or X etc.

Now you are ready to run this program. Type RUN 'DCV', or press the RUN key. If you have entered the program correctly, the HP 3421A will close channel 5, measure DC volts and then display the results on the HP-75D.

When you program the HP 3421A with the DCV command, the HP 3421A A/D converter autoranges, compensates for internal offsets (autozero), sets the A/D converter to the 5½ digit mode (17 bits), and makes a precision DC voltage measurement. See the Appendix for more information on HP 3421A commands.

This program can be easily modified to program the HP 3421A to measure and display AC voltage, resistance, temperature, or frequency on any of the appropriate channels that you have in your HP 3421A. All you have to do is change the program codes in line 50 (see the Appendix of this product note for a list of HP 3421A programming codes). For example, you could change line 50 to one of the following lines to program the HP 3421A for a different measurement:

Alternatives for line 50	Comments
50 OUTPUT D:"ACV2"	Measure AC voltage on channel 2
50 OUTPUT D:"TEM7"	Measure thermocouple, channel 7
50 OUTPUT D:"F3CLS6"	Two wire ohms(F3),channel 6(CLS6)
50 OUTPUT D:"F1N3CLS9"	DC volts, 3 ½ digits, channel 9

The program DCV can also be modified to take readings on a number of channels and display the results. To edit the existing file DCV you can simply use the ▲ key and ▼ key to reach the appropriate line. If you had several program files in the HP-75D and the current file is not DCV, then execute an EDIT 'DCV' statement (type EDIT 'DCV' and press RTN)

Let's change the program DCV to measure DC volts on channels 3 through 7. The program changes required to do

\*NOTE: The " and ' quotation marks are equivalent and can be used interchangeably but not mixed on the HP-75D.

this are shown in the program **Listing 15** below. Change the program lines 20, 50, 55, 60, 70 and 75 to the following:

	Comments
10   DCV HP 75/3421A	Name of program
20 DIM V(5)	Dimension the array V
30 CLEAR LOOP	Clear HP-IL
40 D=DEVADDR('E1')	Assign D=HP 3421A address
50 OUTPUT D ; "DCV3-7"	Measure DCV on channel 3-7
55 FOR I=1 TO 5	Set up FOR-NEXT loop
60 ENTER D ; V(I)	Put the results in the array V
70 DISP V(I)	Display the readings
75 NEXT I	Increment I
80 END	End of the program

**Listing 15**

You can bring line 50 (or any other line) to the HP-75D's display by pressing the **[A]** key until line 50 is on the display, or by executing a **FETCH 50** command. Once line 50 is displayed, you can make the changes shown above by using the **[>]** or **[<]** keys to position the cursor at the proper character. Use the **[I/R]** key to toggle between the insert (**[I]**) and the replace (**[R]**) mode.

To add lines 55 and 75, press the **[ATTN]** key to clear the display, enter **55** or **75** and the program line, and press **[RTN]**. If you need more information about editing an existing program, see section 11 of the HP-75D owner's manual.

You can now run the program DCV again (press **[RUN]**). If you have edited the program correctly, the HP 3421A should measure DC voltage on channels 3 through 7 and display the results on the HP-75D.

Another statement that you should add to this program is the **OPTION BASE 1** statement which sets the array subscript to start at 1 (instead of 0). Enter line 15 as shown:

### 15 OPTION BASE 1

You can have the HP-75D renumber the program lines starting at the beginning of the program by executing the **RENUMBER** statement (press **[ATTN]**, type **RENUMBER** and press **[RTN]**). Once you have done this, the new program should look like the following:

```

10 | DCV HP 75/3421A
20 OPTION BASE 1
30 DIM V(5)
40 CLEAR LOOP
50 D=DEVADDR('E1')
60 OUTPUT D ; "DCV3-7"
70 FOR I=1 TO 5
80 ENTER D ; V(I)
90 DISP V(I)
100 NEXT I
110 END

```

**Listing 16**

The HP-75D has saved this program under the filename DCV in its internal memory. To create another program you simply execute another **EDIT filename** statement. See section 3 of the HP-75D owner's manual for information on file operations.

This program can be further modified to program the HP 3421A to take other multi-channel measurements by changing

the **OUTPUT** statement in line 60 (see the Appendix of this product note for a summary of these codes). You may also have to redimension the array V in line 30 and the **FOR-TO** statement in line 70 for a different number of channel measurements. For example, to read AC voltage on channels 2-9, you would change lines 30, 60, and 70 in **Listing 16** to the following:

```

30 DIM V(8)
60 OUTPUT D;"ACV2-9"
70 FOR I=1 to 8

```

## Printers (HP 82162A, HP 2225B)



The addition of an HP-IL printer like the HP 82162A or HP 2225B will allow you to obtain a hard copy of your program, or to print out the results of your measurements. You can assign the first printer device in the loop to be program printer with a **PRINTER IS 'P1'** statement (make sure that you have executed an **ASSIGN LOOP** or **AUTOLOOP** statement as described previously in HP-IL Basics).

To verify that your printer is working properly, press the **[ATTN]** key to clear the HP-75D's display, type **PRINTER IS 'P1'** and press **[RTN]**. After you have done this, type **PRINT "TEST"** and press **[RTN]**. The printer should now have printed out the word **TEST**. See the HP-75D manuals for information on the **PRINTER IS** statement.

Let's add one more line to our program DCV to assign the printer, and change line 90 from a **DISP** statement to a **PRINT** statement. Remember, to add or change program lines, execute a **EDIT 'DCV'** statement, then press **[FET]** and enter **25** (or **90**), and press **[RTN]**.

```

25 PRINTER IS 'P1'
90 PRINT "CHANNEL";I+2;" = ",V(I)

```

After you have added lines 25 and 90 to the program DCV, you can renumber the program by typing **RENUMBER** and pressing **[RTN]**. You can also obtain a program listing on your printer by typing **PLIST** and pressing **[RTN]**. If you have edited the program DCV correctly the measurements taken on the HP 3421A should be printed (along with the channel



number) on your HP 2225B (or HP 82162A) printer. A program listing and sample print out (from the HP 2225B) is shown below in **Listing 17**.

```


10 1 DCV HP 75/3421A
20 OPTION BASE 1
30 PRINTER IS 'P1'
40 DIM V(5)
50 CLEAR LOOP
60 D=DEVADDR('E1')
70 OUTPUT D ; "DCV3-7"
80 FOR I=1 TO 5
90 ENTER D ; V(I)
100 PRINT "CHANNEL ";I+2;" = ";V(I)
110 NEXT I
120 END

CHANNEL 3 = .031794
CHANNEL 4 = .064478
CHANNEL 5 = .104563
CHANNEL 6 = .119743
CHANNEL 7 = .138173

```

**Listing 17**

## Mass Storage (HP 82161A, 9114A)



In order to use a mass storage device like the HP 82161A digital cassette drive or the HP 9114A disc drive, you must first initialize the medium in the device. For purposes of clarity, this product note will refer to the HP 82161A and the HP 9114A with the same device code (:M1). The mass storage commands (e.g. COPY, ASSIGN, INITIALIZE) executed on the HP-75D will work with either the HP 82161A or the HP 9114A.

To initialize the mass storage device, type **INITIALIZE 'M1'** and press **RTN**. This will initialize the HP 82161A cassette for 128 file entries and the HP 9114A disc drive for 616 file entries. The directory size that you specify may depend on whether you'll have many short files or a few long files. You can specify the number of file entries by typing:

**INITIALIZE 'M1',number of file entries**

You save programs on the mass storage device by typing:  
**COPY 'file name' TO 'file name:M1'**

For example, you can save the file DCV by typing **COPY 'DCV' TO 'DCV:M1'** and pressing **RTN**. To transfer a file from the mass storage device to the HP-75D, type **COPY 'DCV:M1' TO 'DCV'** and press **RTN**. If you have two mass storage devices connected to the HP-75D, you can transfer a file from the first mass storage device to the second mass storage device by typing **COPY 'filename:M1' TO 'filename:M2'** and pressing **RTN**.

To save data on the cassette (or disc), use the **ASSIGN #** and **PRINT #** statements to store data internally in a data file in the HP-75D's memory, and then use the **COPY** statement to store the data file on your mass storage device. The **ASSIGN #** statement is used to create new data files in the HP-75D and to make existing data files available for program input and output. The **PRINT #** statement is used to print values to a particular data file opened by the **ASSIGN #** statement.

## Reading Storage

In many data acquisition applications you may need to know approximately how many HP 3421A readings the HP-75D can store on the cassette (or disc). The HP-75D stores data in a data file by line number; every time it executes a **PRINT #** statement, it increments the current line number of the data file, and stores the data. For storing data in a data file, the HP-75D:

- uses 9 bytes to store a noninteger (e.g. HP 3421A reading)
- uses 5 bytes per line for line number etc.
- uses 2 bytes plus 1 byte per character to store a string (e.g. current time of day)
- maximum of 253 bytes per data line (not including line number).

The maximum number of HP 3421A readings that you can store on the cassette or disc is shown below (assuming 15,000 byte data files). Since the HP-75D must store the data internally and then copy the data file to the mass storage device, the maximum size of your data files is limited by the amount of available memory in the HP-75D. If you store only data on your mass storage device, and stored 28 readings per data line ( $28 * 9 = 252$  bytes), then to store 28 readings plus the line number should take about 257 bytes ( $252 + 5$ ). See the Appendix for an explanation of how these values were determined.

	HP 82161A	HP 9114A
Number of Data Files	9	42
Available Records	508	2456
Available Bytes	130,048	628,736
Maximum Number of Readings	14,168	68,500

You normally will want to store the time that the data was taken, and typically will not store 28 readings per data line. Let's look at an example to calculate the typical number of HP 3421A readings you can store on the cassette (or disc). If you store the time on one data line, and then store 10 HP 3421A readings on the next line, you will use 15 bytes to store the time and 95 bytes to store the data (there is also an additional 15 bytes of overhead for filename etc.), a total of

## Data Logging



Data logging consists of a sequence of measurements (e.g. temperature) repeated at specific time intervals and either saved on a mass storage device or a printer or both. The combination of a HP 3421A, 75D, 2225B, and either the HP 82161A or the HP 9114A, is a powerful data logger that is well suited for either portable or nonportable applications.

Let's start out with a simple example to understand the structure of a data logging program, and then modify the program to see how you could adapt this program to fit your application. This program assumes that you have an HP-75D (with I/O ROM), HP 3421A (with an option 020), an HP 2225B (or HP 82162A) printer and an HP 9114A (or HP 82161A) mass storage device. This first example program will be called LOG1, so type **EDIT 'LOG1'**, press **[RTN]**, and enter in the program lines in **Listing 19** (this program assumes that you have initialized the storage media).

This program will measure DC voltage on channels 2 through 9, store the time and results in data file DLOG1, and print out the results. The program will wait for 1 minute after the start of the previous pass and then start another pass.

After 5 passes (i.e. iterations), the program will close the data file DLOG1, copy the data file on the mass storage device (filename DLOG1), and then purge the data file DLOG1 that is in the HP-75D's internal memory to conserve memory space. If you run this program more than once, it will write over the old data file DLOG1.

Now that you have taken and recorded the data, you can access the data in the file DLOG1 on your mass storage device at a later time. To transfer the data file DLOG1 from your mass storage device into the HP-75D, type **COPY 'DLOG1:M1' TO 'DLOG1'**.

You can display the data in the data file DLOG1 by typing **EDIT 'DLOG1'**, pressing **[RTN]**, and using the cursor (**[V]**, **[A]**, **[<]**, **[>]**) keys. You can print out the data by typing **PLIST** and pressing **[RTN]**. The format should look similar to **Listing 20**.

You can change the program to store both the time and the readings on the same line in your data file by deleting line 90 of LOG1, and changing line 140 to **PRINT #1;T\$,V()**.

125 bytes for 10 HP 3421A readings. You could store 10,403 readings (130,048 bytes/12.5 bytes per reading) on the HP 82161A cassette drive or 50,298 readings (628,736 bytes/12.5 bytes per reading) on the HP 9114A disc drive (assuming 15,000 bytes per data file).

To save data on your mass storage device, you need to create the data file and then access it. Let's write a program called STORE1 that takes DC voltage readings from the HP 3421A's channels 5 through 9, in the 5 1/2 digit mode and stores the results on tape (or disc). Type **EDIT 'STORE1'** and press **[RTN]**, type **AUTO** and press **[RTN]** and enter the program in **Listing 18**.

	Comments
10 STORE1 HP 75/3421A	Name of the program
20 OPTION BASE 1 @ CLEAR LOOP	Array starts at 1; Clear HP-IL
30 DIM V(5)	Dimension array V
40 ASSIGN # 1 TO 'DATA1'	Create data file DATA1
50 D=DEVADDR('E1')	Set D to HP 3421A address
60 OUTPUT D : "DCV5-9"	Measure DC volts, channels 5-9
70 FOR I=1 TO 5	FOR NEXT Loop to enter the
80 ENTER D : V(I)	data into the array V
90 NEXT I	
100 PRINT # 1 : V()	Store in data file DATA1
110 ASSIGN # 1 TO *	Close data file DATA1
120 COPY 'DATA1' TO 'DATA1:M1'	Copies DATA1 to the mass
130 END	storage device (same file name)

### Listing 18

The @ concatenation character in line 20 joins statements, which enables you to enter more than one statement in a program line. See the HP-75D manuals for more information on the @ character.

If you have properly connected the HP 3421A, 82161A or 9114A, and the HP-75D; and you have entered the program correctly, this program should create a data file, take 5 readings from the HP 3421A, and store the results in the file DATA1 on your mass storage device. Press **[RUN]** or type **RUN 'STORE1'** and press **[RTN]**.

For some data acquisition applications, you may want to record the data for a period of time and then analyze it later. This program has stored the data internally in data file DATA1 and externally on the mass storage device in data file DATA1. You may want to purge the internal data file to conserve memory space by typing **PURGE 'DATA1'** and pressing **[RTN]**. If you have the file DATA1 in the HP-75D's internal memory, you can display the data points by typing **EDIT 'DATA1'** pressing **[RTN]**, and then use the cursor (**[V]**, **[A]**, **[<]**, **[>]**) keys to display the data points. If you have purged the file from your internal memory, then execute a **COPY 'DATA1:M1' TO 'DATA1'** statement.

Save this program on your mass storage device by typing **COPY 'STORE1' TO 'STORE1:M1'** and pressing **[RTN]**.

## File Catalog

You can find out which files you have in the HP-75D's memory by typing **CAT ALL** and pressing **[RTN]**, and then using the **[V]** key to display the next file. You can also find out which files are on your mass storage device by typing **CAT ':M1'** and pressing **[RTN]**, and again using the **[V]** key. To get a print out of these catalogs, type **DISPLAY IS ':P1'** and press **[RTN]**, execute **CAT ':M1'**, (or **CAT ALL**) and use the **[V]** key. You can get back to normal display operation by typing **DISPLAY IS \*** and pressing **[RTN]**.

**Comments**

```

10 LOG1 HP 75/3421A ----- Name of the program
20 OPTION BASE 1 @ CLEAR LOOP @ T1=1 ----- Set array subscript;clear loop and set counter T1 = 1
30 DIM V(8),T$(8) ----- Dimension variables
40 PRINTER IS ':P1' ----- Set up printer
50 D=DEVADDR(':E1') ----- Set D = address of HP 3421A
60 ASSIGN # 1 TO 'DLOG1' ----- Create data file DLOG1
70 ON TIMER # 1,60 GOTO 80 ----- Timer interrupt every 60 sec. go to line 80 on interrupt
80 T$=TIME$ ----- Read the current time
90 PRINT # 1 ; T$ ----- Transfer time to DLOG1
100 OUTPUT D ; "DCV2-9" ----- DC volts, channels 2-9
110 FOR I=1 TO 8
120 ENTER D ; V(I) ----- Put measurements in array V
130 NEXT I
140 PRINT # 1 ; V(I) ----- Transfer data to DLOG1
150 PRINT " " ----- Print out a blank line
160 PRINT "TIME=";T$ ----- Print out the time
170 FOR I=1 TO 8
180 PRINT "CHANNEL ";I+1;" =";V(I) ----- Print out the results
190 NEXT I
200 T1=T1+1 ----- Increment counter T1
210 IF T1=6 THEN GOTO 230 ----- After 5 passes, exit program
220 GOTO 220 ----- Wait for next timer interrupt go to line 80 on interrupt
230 OFF TIMER # 1 ----- Disable timer interrupt
240 ASSIGN # 1 TO * ----- Close data file DLOG1
250 COPY 'DLOG1' TO 'DLOG1:M1' ----- Copy data file to mass storage device as file DLOG1
260 PURGE 'DLOG1' ----- Purge data file DLOG1 in HP-75D's memory to conserve space.
270 END

```

```

TIME=10:22:14
CHANNEL 2 = .043499
CHANNEL 3 = .073944
CHANNEL 4 = .094111
CHANNEL 5 = .13528
CHANNEL 6 = .049814
CHANNEL 7 = .192846
CHANNEL 8 = .255024
CHANNEL 9 = .34954

```

Results of the first pass

```

TIME=10:26:14
CHANNEL 2 = .052129
CHANNEL 3 = .118614
CHANNEL 4 = .139324
CHANNEL 5 = .204842
CHANNEL 6 = .186874
CHANNEL 7 = .228587
CHANNEL 8 = .39336
CHANNEL 9 = .48568

```

Results of the fifth pass

**Listing 19**

	<b>Comments</b>
1 DATA '10:22:14'	Time of first pass
2 DATA .043499,.073944,.094111,.13528,.049814,.192846,.255024,.34954	Eight readings(first pass)
3 DATA '10:23:14'	Time of second pass
4 DATA 1.44805,1.43071,1.40899,1.39812,1.37429,1.35462,1.34977,1.36251	Eight readings(second pass)
5 DATA '10:24:14'	.
6 DATA 2.08654,2.0743,2.06856,2.05901,2.04645,2.03014,2.03021,2.03796	.
7 DATA '10:25:14'	.
8 DATA 2.6856,2.67223,2.65197,2.63026,2.59322,2.55489,2.55557,2.55976	.
9 DATA '10:26:14'	Time of fifth pass
10 DATA .052129,.118614,.139324,.204842,.186874,.228587,.39336,.48568	Eight readings(fifth pass)

**Listing 20**

## Comments

```

10 LOG2 HP 75/3421A _____ Name of the program
20 OPTION BASE 1 @ CLEAR LOOP @ T1=1 _____ Set array subscript;clear loop and set counter T1=1
30 DIM V(8),T$(8) _____ Dimension variables
40 PRINTER IS 'P1' _____ Set up printer
50 D=DEVADDR('E1') _____ Set D=address of HP 3421A
60 ON TIMER # 1,300 GOTO 70 _____ Timer interrupt every 5 minutes
70 ASSIGN LOOP @ T$=TIME$ _____ Read the current time, restore device assignments
80 ASSIGN # 1 TO 'DATA' _____ Open data file DATA
90 PRINT # 1 ; T$ _____ Store time in DATA
100 OUTPUT D : "DCV2-9" _____ DC volts, channels 2-9
110 FOR I=1 TO 8
120 ENTER D : V(I) _____ Put measurements in array V
130 NEXT I
140 PRINT # 1 ; V(I) _____ Transfer data to DATA
150 PRINT " " _____ Print out blank line
160 PRINT "TIME=" ; T$ _____ Print out the time
170 FOR I=1 TO 8
180 PRINT "CHANNEL " ; I+1 ; " =" ; V(I) _____ Print out the results
190 NEXT I
200 ASSIGN # 1 TO * _____ Close data file DATA
210 N$='DATA'&STR$(T1)&'M1' _____ N$=name of mass storage data file;for example,when T1=2,N$=DATA2:M1
220 COPY 'DATA' TO N$ _____ Transfer data to N$
230 PURGE 'DATA' _____ Purge data file DATA in HP-75D's memory to conserve space
240 T1=T1+1 _____ Increment counter T1
250 IF T1=25 THEN GOTO 270 _____ After 24 passes, exit program
260 BYE _____ Wait for next timer interrupt, turn HP-75D, 3421A etc. off; on interrupt go to line 70
270 OFF TIMER # 1 _____ Disable timer interrupt
280 END

```

## Listing 21

The data logger program (LOG1) can be easily modified for different measurement functions, different time intervals and number of iterations (or passes). You can modify the program to either print or record only instead of doing both.

Before you modify LOG1, save it on your mass storage device by typing:

**COPY 'LOG1' TO 'LOG1:M1'** and pressing **[RTN]**.

Let's call the new program LOG2, and copy the existing program LOG1 by typing **COPY 'LOG1' TO 'LOG2'** and pressing **[RTN]**. Now you have saved the program LOG1 on tape, LOG1 is still in the HP-75D's memory, and you have created a file LOG2 that you can modify. Type **EDIT 'LOG2'** and press **[RTN]**. Change the program lines that are highlighted in **Listing 21**.

This program will measure DC voltage on channels 2-9 print out the results, store the results on the mass storage device, and repeat the process every 5 minutes for 2 hours.

The data is stored on the mass storage device in 24 files called DATA<sub>X</sub> where X is the number of the pass. For example, data taken on pass number 5 are stored on the mass storage device in file DATA5.

The changes made to the program include changing the timer to a different time between passes (line 60), and changing the number of passes (line 250).

The largest change in this program (LOG2) from LOG1 is that this program stores the data on the mass storage device after every pass (lines 200-220), and then destroys the data file in the HP-75D. You should not run out of HP-75D memory by storing 8 readings per pass and then copying the entire file to your mass storage device after 24 passes. If you changed the program to 30 readings per pass and then wait till after the last pass to transfer the data to the mass storage device (as in LOG1), the data file in the HP-75D will be about 7 Kbytes long! If you increase the number of passes,

you may not even be able to store all the data inside the HP-75D. To calculate this number:

$$24 \times [(30 \times 9) + 16 + 10] \sim 7 \text{ Kbytes}$$

↑ passes     
 ↑ readings per pass     
 ↑ bytes per reading     
 ↑ overhead     
 ↑ bytes for time

Line 210 is one way to assign the name of the mass storage data file. The name of the data file corresponds to the the pass, so that the data taken on pass 3 is stored in data file DATA3, and the data taken on pass 21 is stored in data file DATA21. You can transfer the data from the mass storage device to the HP-75D with a **COPY 'filename:M1' TO 'filename'** statement.

Also note that line 220 changed to a **BYE** statement; this turns the HP-75D off. When the timer has elapsed its 5 minutes, it will turn the HP-75D back on and proceed with the program. When the HP-75D is turned off, it will power down any HP-IL peripherals that will respond to a loop power down command (HP 3421A\*, 82161A, 82162A). The HP-75D will reawaken these peripherals after the timer alarm interrupt wakes it up at the end of 5 minutes.

Using these example data logger programs as a base, you should be able to modify them to fit most of your data logging applications. You can also add lines to the program to take an action based on a limit for a channel being exceeded (see the example programs on the next pages).

\*NOTE: Current HP 3421A's do not turn off their display when entering a loop power down state. HP 3421A's with serial numbers below 2238A03053 turn off their display when they receive a power down command.

## Example Programs

By now you should have an understanding of how to program the HP 3421A with the HP-75D, and how to communicate with other HP-IL peripherals. The following programs are further examples of how to program the HP 3421A with the HP-75D.

**Problem:** Program the HP 3421A to take 16 readings of DC voltage on channels 6-9 (4 readings per channel) and display the average of those 16 readings.

**Solution:**

	Comments
10 OPTION BASE 1 @ CLEAR LOOP @ S=0	Array subscript=1;clear loop; set S = 0
20 DIM A(4,4)	Dimension array A
30 D=DEVADDR('E1')	Set D=address of 3421A
40 FOR I=1 TO 4	Set up FOR-NEXT loop
50 OUTPUT D : "DCV6-9"	DC volts on channels 6-9
60 FOR J=1 TO 4	Set up loop to read data
70 ENTER D : A(I,J)	Enter data into array A
80 S=S+A(I,J)	Add data points to S
90 NEXT J	
100 NEXT I	
110 DISP "AVERAGE=";S/16	Display the average of the readings
120 END	

**Listing 22**

**Problem:** You have two HP 3421A's in the loop, and you need to program the first one to measure DC volts on channels 3-7 and print out the results. The second HP 3421A needs to monitor temperature on channels 2-9 for 10 minutes, and close actuator channel 1 for 30 seconds (turn on a fan) if any channel exceeds 30 degrees C.

**Solution:**

	Comments
10 OPTION BASE 1 @ PRINTER IS 'P1'	Set array subscript;assign printer
20 DIM X(5),V(8)	Dimension arrays X,V
30 D1=DEVADDR('E1')	Set D1=address of first HP 3421A in the HP-IL loop
40 D2=DEVADDR('E2')	Set D2=address of second HP 3421A in the HP-IL loop
50 OUTPUT D1 : "DCV3-7"	DC volts, channels 3-7
60 FOR I=1 TO 5	Set up FOR-NEXT loop
70 ENTER D1 : X(I)	Enter data into array X
80 PRINT "CHANNEL ";I+2;"=";X(I)	Print out the results
90 NEXT I	
100 ; PROGRAM SECOND HP 3421A	Comment line
110 ON TIMER # 1,600 GOTO 230	Exit program after 10 minutes
120 OUTPUT D2 : "TEM2-9"	T type T/C's, channels 2-9
130 FOR I=1 TO 8	Set up FOR-NEXT loop
140 ENTER D2 : V(I)	Enter data into array V
150 IF V(I)>30 THEN GOTO 180	If the temperature is greater than 30, go to line 180
160 NEXT I	
170 WAIT 10 @ GOTO 120	Return to monitoring temp.
180 PRINT "CHANNEL ";I+1;"> 30 DEG.C"	Print channel number>30 deg.
190 OUTPUT D2 : "CLS1"	Close actuator channel 1
200 WAIT 30	Wait 30 seconds
210 OUTPUT D2 : "OPN1"	Open actuator channel 1
220 GOTO 120	Return to monitoring temp.
230 OFF TIMER # 1	Turn off timer, exit program
240 END	

**Listing 23**

**Problem:** You need to benchmark the HP-75D/3421A combination for reading rates (with and without transfer times) for various modes of operation.

**Solution:**

	Comments
10 OPTION BASE 1 @ PRINTER IS ':P1'	Set array subscript; assign printer
20 DIM A(30)	Dimension array A
30 D=DEVADDR(':E1')	D=HP 3421A address
40 ASSIGN # 1 TO 'DBURST'	
50 OUTPUT D : "T0LS2-9,2-9,2-9,2-7:F1N3R0Z0"	T0 Wait for trigger;enable channel list,reading storage
60 T1=TIME	Read the current time
70 OUTPUT D : "T3"	Trigger the HP 3421A
80 T2=TIME	Read the time after
90 FOR I=1 TO 30	measurements have
100 ENTER D : A(I)	been completed
110 NEXT I	Set up FOR-NEXT loop to
120 T3=TIME	transfer the data from the
130 PRINT # 1 ; A()	HP 3421A's internal buffer
140 PRINT "READING TIME=";30/(T2-T1);"RDGS/SEC"	to the HP-75D.
150 PRINT "READING PLUS TRANSFER TIME=";30/(T3-T1);"RDGS/SEC"	Read the time after the
160 END	measurement and transfer
	is completed
	LS2-9,2-9,2-9,2-7 Load channel list
	F1 DC volts
	N3 3½ digits
	R0 Fixed range
	Z0 Auto Zero off

```

READING TIME= 23.8095238095 RDGS/SEC
READING PLUS TRANSFER TIME= 3.21784833208 RDGS/SEC

```

**Listing 24**

You can modify this program for a different number of digits of resolution, auto zero on or off, different function, or different range by changing the HP 3421A programming codes in line 40. This program will take readings on 30 channels; to take 30 readings on a single channel, change line 40 to the following: **40 OUTPUT :D;"TOLS4-4:F1N3R0Z0"**

Here are some typical numbers for the HP 3421A controlled by the HP-75D: (these numbers do not subtract the time it takes to read the HP-75D's internal clock, so the actual reading rates should be slightly faster)

HP-75D/3421A			
DC Volts*		Auto Zero	
		OFF	ON
Different Channels	3½ digits	23/3	20/3
	4½ digits	17/3	12/2
	5½ digits	3/1	2/1
Same Channel	3½ digits	34/3	28/3
	4½ digits	22/3	15/2
	5½ digits	4/1	2/1

\* The first number is the reading rate (per second) into the HP 3421A's internal buffer; the second number is the reading rate (per second) including reading time plus transfer time to the HP 75D.

You can read the data from the file DBURST and print out all 30 readings with the program in **Listing 25**.

```

10 OPTION BASE 1 @ PRINTER IS ':P1'
20 DIM R(30)
30 ASSIGN # 1 TO 'DBURST'
40 READ # 1 ; R()
50 FOR I=1 TO 30
50 PRINT "READING #";I;" =";R(I)
70 NEXT I
80 ASSIGN # 1 TO *
90 END

```

**Listing 25**

### Thermistor, Thermocouple and RTD Linearization

Thermistor and thermocouple linearization programs are found on pages 76 and 78 of the HP 3421A manual; you should be able to run these programs by changing the OUTPUT 901 and ENTER 901 statements to OUTPUT D and ENTER D and adding an D=DEVADDR(':E1') statement to the beginning of the program. Linearization for a 100 Ohm Platinum RTD (alpha=0.00385) is shown in this example:

```

10 D=DEVADDR(':E1')
20 OUTPUT D ;"FW02"
30 ENTER D ; C
40 W=C/100
50 IF W<1 THEN 100
60 P1=3367.82144088
70 P2=13065764.8633
80 P3=-1723543.60565
90 P=P1-SQR(P2+P3*W) @ GOTO 150
100 P0=-241.996759172
110 P1=222.560617915
120 P2=25.2488238815
130 P3=-5.81268262546
140 P=P0+W*(P1+W*(P2+W*P3))
150 DISP P;"DEGREES C"
160 END

```

# IV. Appendix

## HP 3421A Programming Codes

The HP 3421A command set consists of standard commands and advanced commands. Each standard command performs a complete measurement or function while two or more advanced commands are generally required to perform a measurement. Refer to the blue pages in the center of the 3421A Operating, Programming, and Configuration Manual for more information and program examples on each command.

[ ] means optional channel or bit list

< > means mandatory channel or bit list

### Standard Commands

**DCV [x,y,...z]** DC Volts. Sets the voltmeter to DCV (F1), Autorange (RA1), Autozero on (Z1), 5½ digits resolution (N5). If no channel list is sent, the channel list is not changed but software single trigger is executed (T2). If channel list is sent, they are loaded in order received then a reading is made and stored from each channel in sequence (T3). DCV always opens the last channel before closing the next channel in the list. It exits with the last channel in the list closed unless no numbers were received then it exits with the channels in the same state they were in prior to the command. When the 3421A is addressed to talk, all readings will be sent in the sequence they were taken.

**ACV [x,y,...z]** Same as DCV but for AC Volts (F2) and 4½ digit resolution (N4).

**TWO [x,y,...z]** Same as for DCV but for 2-wire Ohms (F3) measurements.

**FWO [x,y,...z]** Same as for DCV except for 4-wire Ohms (F4) measurements. Channels are automatically paired with x + 10, y + 10, ... z + 10 unless x,y,...z are between 20 and 29 in which case they are paired with x - 20, y - 20, ... z - 20. Pairs are closed simultaneously.

**TEM [x,y,...z]** Same as for DCV but for temperature measurements (F6) and does a software compensated T-type thermocouple conversion. Result returned is in Degrees C. TEM will take a REF temperature measurement on the 44462A assembly in the lowest numbered slot if no channel numbers are sent.

**REF [x]** Measures the temperature of the REFERENCE junction (F5) on 44462A assembly where channel 'x' is. If 'x' is not sent, then defaults to assembly where a multiplexer channel is closed. If no channel is closed, then selects 44462A assembly in lowest numbered slot. Result returned in Degrees C.

**FRQ [x,y,...z]** Measures FREQUENCY (F7) with a 1 second gate time (GO), 5½ digits resolution (N5). If no channel numbers are sent, the channel list is not changed and no channels are opened or closed. A software single trigger (T2) is executed. If channel list is sent they are loaded in the order received and a

reading is made and stored from each channel in sequence. When addressed to talk, all readings will be sent in the sequence they were taken.

**TOT [x]**

TOTALizes events (F7) up to a maximum count of 65,535. If channel 'x' is sent, all channels will be opened before closing channel 'x'. The counter will be zeroed and then starts totalizing. If 'x' is not sent, then the counter is zeroed and starts totalizing without changing channels. Channel 'x' will remain closed until another command opens it. When the 3421A is addressed to talk, it will send out the current subtotal without disrupting the counter. NOTE: if a TRIGGER command is received, the TOT is aborted and a frequency reading is made.

**CLS <x>**

Close Single channel 'x'. The 3421A first identifies the type of channel at 'x' and then if 'x' is:

an Actuator - closes channel x possible actuator channel numbers are: 00,01,10,11,20,21.

a Digital Output Bit - closes it.

a Multiplexer - opens all multiplexer relays and closes channel 'x'.

**CLP <x>**

CLose a Pair of channels. The 3421A will open all multiplexer relays and then close channels x and x + 10. If x ≥ 20, then x and x - 20 will be closed. If either x or its pair is not a multiplexer channel, then no channels are closed or opened and an error is generated.

**OPN[x]**

Open channel(s). If channel 'x' is not sent, then the OPN command will open all channels digital outputs, actuators, and multiplexers. If 'x' is sent, the 3421A identifies the channels and if 'x' is:

an Actuator - opens it.

a Digital Output Bit - clears bit 'x'

a Multiplexer - opens it. This includes channels closed by the UC command.

If 'x' was closed as a pair (i.e., CLPx) then its pair will be opened also.

**REDi**

READs the digital input byte from slot i and replies with a decimal number from 0 to 255. This decimal number represents the values of the bits that were set. Must be an integer value. No decimal points or extra digits allowed.

**WRTi,[ab]c**

WRITe the decimal value [ab]c to slot i. The value [ab]c ≤ 255. If a and/or b not received then the 3421A assumes leading zeros. Must be an integer value. No decimal points or extra digits allowed.

**BIT <x> [y,...z]**

Reads the digital input bits (up to 30 in the bit list) and sends +0.000E+0 if the bit is low or +1.000E+01 if the bit is high. Invalid bit numbers are 08,09,18,19,28, and 29. Also configures for 3-digit readings.

## Advanced Commands

Function, Command	- 1 Range Codes (RA1 = Autorange on, RAO = off)								
	R-1	R0	R1	R2	R3	R4	R5	R6	R7
All functions off, F0	x	x	x	x	x	x	x	x	x
DC Volts, F1	.3V	3V	30V	300V	—	—	—	—	—
AC Volts, F2	—	3V	30V	—	—	—	—	—	—
2-Wire Ohms, F3	—	—	—	300Ω	3kΩ	30kΩ	300kΩ	3MΩ	30MΩ
4-Wire Ohms, F4	—	—	—	300Ω	3kΩ	30kΩ	300kΩ	3MΩ	30MΩ
REFErence, F5	x	x	x	x	x	x	x	x	x
TEMperature, F6	x	x	x	x	x	x	x	x	x
FREquency, F7	Gate time: G-1 = .1S, GO = 1S, G1 = 10S								

**Voltmeter Autzero:** Z0 = Auto-zero off, Z1 = Auto-zero on.

**Number of digits of resolution;** N3 = 3<sup>1/2</sup>, N4 = 4<sup>1/2</sup>, N5 = 5<sup>1/2</sup>

**Trigger:** T0 = Hold trigger and enable channel list scan  
 T1 = Internal trigger  
 T2 = Software single trigger  
 T3 = Triggers measurements from channel list and stores readings  
 DTa = Digital Trigger, does a T3 when digital input bit 'a' goes low.

**AN**[decimal value 0-255] **ANd** mask, used with MN command.

**C**[cal value] **Calibrate**, see calibration procedure.

**DC**[slot number, decimal integer value 0-255] **Digital Clear** specified output bits.

**DS**[slot number, decimal integer value 0-255] **Digital Set** specified output bits.

**DN**[number 0-29] **Display Number**, note: send DN alone (with terminator) to turn off mode.

**LS**<channel list> **Load Single** channels into channel list.

**LP**<channel list> **Load channel Pairs** into channel list.

**M**[decimal integer number] **Set SRQ Mask**.

**MH**<digital bit> **Monitor digital input bit and SRQ interrupt** when bit goes high. Exponents illegal; ignores everything after decimal point.

**ML**<digital bit> **Monitor digital input bit and SRQ interrupt** when bit goes low. Exponents illegal; ignores everything after decimal point.

**MN**<slot number> **Monitor slot and compare to AN mask and XR mask**. SRQ interrupt when result = 0. Must be integer value.

**RL** **Read channel List**.

**RS** **ReSet**.

**SI**<0 or 1> **SI0** = Initialize channel list pointer to beginning of list. **SI1** = opens channel and closes next channel in list. Must be an integer value.

**SR** **Read Status Registers**.

**UC**<channel number> **Unconditionally Close** specified channel.

**XR**[decimal integer value 0-255] **eXclusive-oR** mask, used with the MN command.

## HP-IL Accessories

### Other Products Available

HP 82168A Acoustic Coupler (Modem)  
 HP 82169A HP-IL/HP-IB Converter  
 HP 82164A HP-IL/RS232 Interface  
 HP 82166A HP-IL to GPIO Converter  
 HP 11341A/B HP 3421A Carrying Case  
 HP 3468A HP-IL Digital Multimeter  
 HP 11096B 500 MHz RF Probe

#### HP-IL Cables

HP 11340A 20 Meter HP-IL Cable  
 HP 11340B 50 Meter HP-IL Cable  
 HP 11340C 100 Meter HP-IL Cable  
 HP 82167A 0.5 Meter HP-IL Cable  
 HP 82167B 1 Meter HP-IL Cable  
 HP 82167D 5 Meter HP-IL Cable

#### HP 3421A Accessories

HP 34118A Test Lead Kit  
 HP 44468A 3421A Data Acquisition Pac for 41CV to allow fast programming of the 3421A Acquisition/Control Unit from the HP 41C or HP 41CV. (One 44468A comes with each 3421A, Option 541.)

#### Printer Accessories

HP 82045A Extra printer paper for 82162A  
 HP 92261N 2500 sheets of paper for HP 2225B  
 HP 92261A Inkjet Print Head Cartridge (HP 2225B)

#### Chargers

HP 82059D 120 Volt charger (for 82161A, 82162A, etc.)  
 HP 82066B 220 Volt European charger (for 82161A, 82162A, etc.)

#### HP 41CV Accessories

HP 82153A Wand  
 HP 82170A Quad memory Module for 41CV  
 HP 82182A Time Module for 41CV  
 HP 82160A HP-IL Interface for 41C/CV

#### Mass Storage Accessories

HP 82176A Extra cassettes for 82161A  
 HP 92192A Extra discs for HP 9114A

#### HP-71B Accessories

HP 82420A 4K Memory Modules  
 HP 82400A Card Reader  
 HP 82401A HP-IL Interface

#### HP-75D Accessories

HP 82700A 8K Memory Module  
 00075-15001 I/O ROM



## Manual Part Numbers

		HP Part Number
HP 3421A	Operating, Programming and Configuration Service	03421-90010
HP-71B	Owner's Reference	03421-90008 00071-90001 00071-90010
HP 82401A	HP-IL Interface Owner's Manual (for the HP-71B)	82401-90001
HP-75D	Owner's Reference I/O ROM	00075-90001 00075-90004 00075-90243
HP-9114A	Operator's	09114-90000
HP 82161A	Owner's	82161-90002
HP 2225B	Owner's	02225-90002
HP 82162A	Owner's	82161-90001
HP-IL System:	An Introductory Guide to the Hewlett-Packard Interface Loop	5955-9425

## Further Information on Mass Storage Devices

The INITIALIZE statement prepares a mass storage medium for storing information. Each medium must be initialized at least once to establish directory space on the medium.

The "directory space" created at initialization (uses the Logical Interface Format - LIF) defines the number of "file entries" that is to be provided in the directory - that is, the number of files that the medium can store. An integral number of "records", where each "record" contains 8 "file entries", is allocated for the "directory space". In addition, two records outside of the "directory" are reserved on the medium for system use (volume information). All remaining space on the medium is available for information storage.

The INITIALIZE statement allocates "directory space" by rounding the number of "file entries" specified up to the nearest multiple of 8, and, then, dividing by 8 to determine the number of "records" that are set aside for the "directory".

Each directory "file entry" consists of 32 bytes, and includes information such as file name, starting address, extensions, file type, time of creation, number of registers, number of bytes, and security or privacy information.

The smallest directory size is 1 (although 8 "file entries" will be provided). The largest size depends on the number of "records" available on the medium:

number of "file entries"  $\leq 8/9$  (number of records - 2)

Note that a larger directory leaves less space for storing information than a smaller directory does. If the directory size is not specified, the number of directory "records" is automatically set to 1/32 the number of "records" available on the medium - the number of directory "file entries" is 8 times that number. In determining directory size, any fractional portions are ignored.

### INITIALIZE Examples:

1. The HP 9114A Disc Drive has a total capacity of 2464 sectors (records) on the disc for Series 70 computers. The

maximum directory size is 2118 "file entries". If a directory size is not specified when the disc is initialized (or if the maximum number of file entries is exceeded), the directory size is set to 616 "file entries".

If you specify 1 for the number of "file entries", one "record" will be set aside for the "directory". This directory size leaves 2461 sectors (2464-1-2) available for storing 630,016 bytes of information.

If you specify 42 for the number of "file entries", six "records" will be set aside for the "directory". This directory size leaves 2456 records (2464-6-2) available for storing 628,736 bytes of information.

2. The HP 82161A Digital Cassette Drive\* has a total capacity of 512 records on a cassette tape. The maximum directory size is 453 "file entries". If a directory is not specified when the cassette is initialized, the directory size is set to 128 "file entries".

If you specify 1 for the number of "file entries", one "record" will be set aside for the "directory". This directory size leaves 509 records (512-1-2) available for storing 130,304 bytes of information.

If you specify 9 for the number of "file entries", two "records" will be set aside for the "directory". This directory size leaves 508 records (512-2-2) available for storing 130,048 bytes of information.

### Capacity for HP 82161A Digital Cassette Drive Formatted Using HP Series 40 and 70 Computers

Bytes/Record: 256

Records/Track: 256

Tracks/Cartridge: 2

Bytes/Drive: 128 Kbytes

### Capacity for HP 9114A Disc Drive Formatted Using HP Series 40 and 70 Computers

Bytes/Sector: 256

Sectors/Track: 16

Tracks/Surface: 80

Surfaces/Disc: 2

Bytes/Drive: 630 Kbytes\*\*

### Important Facts:

1 Kilobyte = 1024(2<sup>10</sup>) bytes

1 physical record = 1 sector = 256 bytes for Series 40 and 70 controllers.

1 register = 8 bytes for Series 40 and 70 controllers.

\* When you are using the HP 82161A for Data Logging, the HP-IL POWER-UP command will cause the cassette to completely rewind each time; the required time between data logging scans increases as you start to fill up the cassette.

\*\* On the HP 9114A tracks are spared for data reliability. The spared tracks are used as replacements for bad sections of the disc to ensure reliable data storage. As a result, the actual usable space on the HP 9114A when using Series 40 or 70 computers is 630 Kbytes.


---

## Other Data Acquisition Products


### Data Acquisition Front Ends




**HP 3421A**



**HP 3497A**



**HP 6940/42/44**



**HP 2250**

Contact your local Hewlett Packard sales office for more information on these products.

## Controllers



HP 41CV




HP-71B




HP-75D



HP-85B




HP Series 200




HP 1000

# HP 3421A Data Acquisition/Control Unit

## Standard




- 5<sup>1</sup>/<sub>2</sub> Digit A/D
- VDC, VAC, Ohms
- Counter/Totalizer
- Type T Thermocouple Linearization
- Electronic Calibration
- HP-IL
- Battery Operation
- 30 Reading Storage
- 30 Channel Sequence List
- High Level Instructions




- HP-IL (standard)
- HP-IB (optional)


## Optional



Multiplexer/Actuator Assembly  
(Thermocouple Reference Junction)



Breadboard Assembly



Digital I/O Assembly

See the HP 3421A data sheet (publication number 5953-6975) for complete descriptions and specifications of the HP 3421A Data Acquisition/Control Unit, or call your local HP sales office listed in the telephone directory white pages.